

The background of the cover is a stylized savanna landscape. At the top, there are silhouettes of acacia trees against a sky with horizontal bands of green, orange, and blue. Below the trees is a solid brown ground area. A silhouette of a man, Harry, is shown swinging on a vine from the right side of the frame. He is wearing a hat and a shirt, and his legs are spread out. The title 'PROGRAMMING PITFALL HARRY™' is printed in white, bold, sans-serif capital letters on the left side of the brown ground area.

**PROGRAMMING
PITFALL
HARRY™**

by
David
Crane

ACTIVISION

FOR COMMODORE 64™ OWNERS

Hi! I'm David Crane, senior designer at Activision.

As a Commodore 64 owner, you are probably beginning to write your own programs (or maybe you are quite proficient at programming). Activision has six outstanding entertainment software titles available on disk and cartridge for your C-64, including my game, Pitfall!. I've decided to take you behind the scenes of Pitfall! by sharing a program that will let you put Pitfall Harry on your screen.

This program is divided into three main parts: Pitfall Harry's graphics in DATA statements from 1000 to 1240, an interrupt handler written in assembly language in DATA statements from 900 to 970, and BASIC statements from 20 to 730 to set up the display. It is a simple program, but one which allows a great deal of expansion for you, the programmer.

As your first step toward controlling Pitfall Harry, set up your COMMODORE 64 with disk or cassette and type in the following program, being careful to include all punctuation as noted. When finished, save the program so you can use it later or modify it as you see fit.

```
20 V=53248:REM VIC ADDRESS
30 REM READ INTERRUPT HANDLER INTO
35 REM CASSETTE BUFFER
40 A=832
50 READ B:IF B=999 THEN 100
60 POKE A,B
70 A=A+1
80 GO TO 50
100 REM LOAD PITFALL HARRY'S RUNNING
110 REM PICTURES FROM DATA STATEMENTS
120 REM INTO RAM AT ADDRESS 8192
130 FORS=8192 TO 8192+383
140 READ A:POKE S,A
150 NEXT S
200 REM SET UP SPRITE 0 FOR HARRY
210 REM COLORS ...
220 POKE V+37,10:POKE V+38,9:POKE V+39,5
230 REM MULTI-COLOR MODE ...
240 POKE V+28,1
250 REM SPRITE ENABLE ...
260 POKE V+21,1
270 REM VERTICAL POSITION ...
280 POKE V+1,104
290 REM SPRITE POINTER ...
300 POKE 2040,128
310 REM DOUBLE SIZE X AND Y
320 POKE V+23,1:POKE V+29,1
330 REM LET HARRY PASS BEHIND CHARACTERS
340 POKE V+27,1
400 REM PRINT SOMETHING FOR HARRY TO
410 REM RUN ON ...
420 PRINT CHR$(147);:REM CLEAR SCREEN
430 FOR I=1 TO 12:PRINT:NEXT I
440 FOR I=1 TO 5:PRINT "PITFALL!":NEXT I
500 REM START INTERRUPT RUNNING
510 POKE 788,66:POKE 789,3
520 POKE V+18,240
600 REM KEY HIT MAKES HARRY STAND STILL
610 GET A$:IF A$="" THEN 610
620 FOR I=1 TO 200:POKE 832,48:NEXT I
630 GO TO 610
700 REM STOP INTERRUPT—THIS RESTORES
710 REM SYSTEM TO NORMAL
720 POKE 788,49:POKE 789,234
730 STOP
900 REM ASSEMBLY LANGUAGE IRQ
910 DATA 0,24,206,64,3,16,5,169
920 DATA 39,141,64,3,173,64,3,41
930 DATA 1,208,3,238,65,3,173,65
940 DATA 3,10,141,0,208,42,141,16
950 DATA 208,173,64,3,74,74,74
960 DATA 24,105,128,141,248,7
970 DATA 76,49,234,999
1000 REM PITFALL HARRY'S GRAPHICS
1005 DATA 0,60,0,0,52,0,0
1010 DATA 20,0,0,16,0,0,40,0
1015 DATA 0,40,64,0,170,128,2,170
1020 DATA 0,1,40,0,0,60,0,0
```

```

1025 DATA 42,128,3,40,128,15,160,192
1030 DATA 0,0,240,0,0,0,0,0
1035 DATA 0,0,0,0,0,0,0,0
1040 DATA 0,0,0,0,0,0,0,0
1045 DATA 0,60,0,0,52,0,0,20
1050 DATA 0,0,16,0,0,40,0,0
1055 DATA 40,0,0,168,0,0,168,64
1060 DATA 0,170,128,0,166,0,0,60
1065 DATA 0,0,40,0,0,42,0,0
1070 DATA 170,128,0,162,128,2,128,128
1075 DATA 2,0,192,15,0,240,12,0
1080 DATA 0,3,0,0,0,0,0,0
1085 DATA 0,60,0,0,52,0,0,20
1090 DATA 0,0,16,0,0,40,0,0
1095 DATA 40,0,0,40,0,0,40,0
1100 DATA 0,41,0,0,41,0,0,60
1105 DATA 0,0,40,0,0,42,0,0
1110 DATA 42,128,0,34,128,0,160,128
1115 DATA 0,160,128,0,131,0,0,128
1120 DATA 192,0,192,0,0,48,0,0
1125 DATA 0,60,0,0,52,0,0,20
1130 DATA 0,0,16,0,0,40,0,0
1135 DATA 40,0,0,40,0,0,40,0
1140 DATA 0,40,0,0,41,0,0,41
1145 DATA 0,0,60,0,0,40,0,0
1150 DATA 42,0,0,10,128,0,10,128
1155 DATA 0,234,0,0,200,0,0,200
1160 DATA 0,0,12,0,0,15,0,0
1165 DATA 0,60,0,0,52,0,0,20
1170 DATA 0,0,16,0,0,40,0,0
1175 DATA 40,0,0,168,0,0,168,64
1180 DATA 0,170,64,0,166,0,0,40
1185 DATA 0,0,40,0,0,40,0,0
1190 DATA 42,0,0,42,0,0,34,0
1195 DATA 3,226,0,3,3,0,3,0
1200 DATA 240,0,0,192,0,0,0,0
1205 DATA 0,60,0,0,52,0,0,20
1210 DATA 0,0,16,0,0,40,0,0
1215 DATA 40,0,0,40,64,0,42,128
1220 DATA 0,42,0,0,40,0,0,40
1225 DATA 0,0,60,0,0,40,0,0
1230 DATA 40,0,0,40,0,0,40,0
1235 DATA 0,40,0,0,44,0,0,47
1240 DATA 0,0,48,0,0,60,0,0

```

With the program now in memory, type RUN. After a few seconds Pitfall Harry should appear running across the screen on a line made up of the word PITFALL!. Now, amaze your friends... stop the program (using the RUN/STOP key) and Harry keeps running! Nothing can stop Pitfall Harry, not even stopping the program. Type LIST and watch the entire program scroll up the screen without causing Harry to miss a step. How can this be?

Lines 900-970 contain an assembly language program which makes Harry run. This program is moved into memory by lines 30-80, and then lines 500-520 cause this program to be executed every 1/60th of a second, regardless of what BASIC is doing. Thus, Harry will run even when the BASIC program is stopped.

AMAZE YOUR FRIENDS WITH THESE CUSTOMIZED CHANGES

To make any changes, hit the RUN/STOP key on your C-64, then LIST the appropriate line number. Then, position the cursor over the number(s) you wish to change and experiment with. After making the change, hit the RETURN key and type RUN.

CHANGE HARRY'S SPEED

The assembly language portion of the program begins with the third number in line 910 and continues to the 999 in line 970. Any modifications in this section will change the way Harry runs. For example, changing the first number in line 930 from 1 to 3 slows Harry down, changing from 1 to 0 speeds Harry up. Caution is advised, however. Any mistakes made in this section can destroy your program, so these changes should be made with care by advanced programmers only.

MAKE HARRY RUN ON TOP OF BUILDINGS

Greater modification can be made in lines 400-440. This section currently writes a line of PITFALL!'s for Harry to run on. A creative programmer could use the COMMODORE 64 graphics character set to draw a more interesting scene, such as a city skyline or a jungle path. Try something like that and your program will look better than mine.

MAKE HARRY'S SHIRT RED...AND MORE!

Lines 200-440 are easily modified to change the display. For example, line 220 controls Harry's hair, face and clothes color; changing the last number in line 280 will change Harry's vertical position on the screen; or POKEing 0 instead of 1 in line 340, makes Harry show up in front of screen characters rather than behind them.

Finally, the most extensive modifications can be made in Harry's running pictures. Lines 1000-1240 represent 5 running animation pictures and 1 standing picture of 64 bytes (numbers) each arranged as explained in your COMMODORE 64 Programmer's Reference Guide.

Have fun, try these modifications and some of your own, and if you create something you are very proud of, send me a disk, I'd love to see it. All materials sent to Activision become the property of Activision and cannot be returned, but if we like something very much, we will contact you. There may be a future for YOU in video game design! Above all, enjoy programming and watch for Activision's latest software for use with your COMMODORE 64.

Sincerely,

David Crane



Write to:
Commodore 64 Club
Activision Inc.
P.O. Box 7287
Mountain View, CA 94039
© 1984, Activision Inc.